```
ClearAll["Global`*"]
```

Hálózatanalízis frekvenciatartományban

Bevezetés

A dolgozatban elsősorban folytonos idejű hálózatok frekvenciatartománybeli analízisével foglalkozik, az átviteli függvények lehetséges ábrázolásaival, valamint folytonos idejű hálózatok diszkrét idejű szimulálásával.

Függvények ismertetése

Külső csomagok

Most egy *Mathematica*-hoz telepíthető külső csomag -TechPackages`AnalogCircuits` -segítségével ábrázolunk hálózatokat. A programcsomagban az egyes komponensek rajza a *Mathematica* beépített grafikai primitívjeivel van elkészítve, és ezek egy mátrixba vannak foglalva.

```
<< TechPackages`AnalogCircuits`
```

Az elemek nevének a beviteléhez palettát nyit. Fő utasítása **MakeDiagram**. A hálózati elemeket a kívánt elhelyezkedésnek megfelelően kell egy mátrixba bevinni. majd a **MakeDiagram** utasítással kirajzoltathatók.

```
<< TechPackages`DiscreteCircuits`
```

Ez előző mintájára készült saját csomag. Diszkrét idejű hálózatok komponenseinek primitívjét tartalmazza, így lehetővé teszi diszkrét idejű hálózat ábrázolását *Mathematica*-ban. Kirajzoló utasítása **MakeDiscreteDiagramm**.

Saját függvények

A függvények a Bode- illetve Nyquist-diagrammok, valamint a Pólus-zérus elrendezés ábrázolásához készültek. Most csak a paraméterezésüket mutatjuk be röviden, jelentésük tárgyalására később térünk rá.
Az első két függvény sajátossága, hogy a többi függvény használja őket.

A függvények működéséhez a LogLinearPlot utasítás miatt szükség van az alább *Mathematica* csomagra.

```
<< Graphics`Graphics`
```

Poles

```
Poles[W_, s_] :=
  Last /@ Flatten[NSolve[Denominator[W // Together] == 0, s]];
```

Ez a függvény a bemeneten kapott átviteli függvény, vagy karakterisztika nevezőiknek zérushelyeit tartalmazó listával tér vissza. Numerikus értéket ad vissza, ezért numerikusan kifejezett W függvényt vár első argumentumkén. A második argumentum a változó. Ez többnyire s, z ,jω.

Zeros

```
Zeros[W_, s_] :=
  Last /@ Flatten[NSolve[Numerator[W // Together] == 0, s]];
```

Mint az előző, de ez a számláló zérushelyeit keresi meg.

BodePlot

```
FreqPlotRange[f_, x_] :=
 Module[{rangelist = Map[Abs[#] &, Select[Chop /@
       Flatten[{Poles[f, x], Zeros[f, x]}], Re[#] ≠ 0 ∨ Im [#] ≠ 0 &]]},
  If[rangelist == {}, {2 π^-2 , 2 π 10}, {2 π^-2 Floor[Min[rangelist]],
    2 π * Ceiling[Max[rangelist]]}]]
```

```
BodePlot[W_, s_, range_Rule, type_Symbol, opts___?OptionQ] :=
 Block[{ω, dom, k = (#1 &), v = (20 Log[10, #1] &), p = (10 Log[10, #1] &),
   scalestyle = {Dashing[{.005, .0075}], GrayLevel[.5]},
   globalopts = {DisplayFunction → Identity, GridLines →
      {{Automatic, scalestyle}, {Automatic, scalestyle}}}},
  If[(FreqRange /. {range}) === Automatic,
   Print[FreqPlotRange[W, s]];
   dom = Insert[FreqPlotRange[W, s], ω, 1],
   dom = Insert[FreqRange /. {range}, ω, 1]];
  ampplot = LogLinearPlot[Evaluate[type[Abs[W /. s → ⅈ ω]]],
    dom, Evaluate[FilterOptions[Plot, globalopts]],
    Evaluate[FilterOptions[Plot, opts]],
    AxesLabel → {"ω", If[type === k, "PU", "dB"]}];
  phaseplot = LogLinearPlot[Evaluate[180 Arg[W /. s → ⅈ ω] / π], dom,
    Evaluate[FilterOptions[Plot, globalopts]], PlotRange → All,
    Evaluate[FilterOptions[Plot, opts]], AxesLabel → {"ω", "ϕ"}];
  Show[GraphicsArray[{{ampplot}, {phaseplot}},
    DisplayFunction → $DisplayFunction], Frame → True,
   PlotLabel → "Bode - diagamm\nAmplitúdó & Fázis"];]
```

Bode diagrammot rajzol. Bemenő paraméteri az átviteli függvény, a függvényben szereplő frekvencia változó. Ezután az ábrázolni kívánt frekvencia intervallumot kell megadni. Lehetséges opció a Freqrange → Automaticis, ilyenkor a pólus és zérushelyek figyelembevétel, automatikusan határozza meg az ábrázolás szempontjából kedvező intervallumot. A következő paraméter az amplitúdó diagrammra vonatkozik. Meghatározható vele, hogy arányt szeretnénk ábrázolni - ezt a *k*-val tehetjük - illetve dB-t. Ez utóbbi további két lehetőséget rejt: *v* a jó paraméter, ha feszültséget ábrázolunk, *p* pedig, ha teljesítményt. A következő paraméter(ek) a megjelenítésre vonatkozó utasítások lehetnek, megegyeznek a **Plot** parancs lehetőségivel.

NyquistPlot

```
NyquistPlot[W_, s_, type_Symbol, opts___?OptionQ] :=
 Module[{nyp, ω, freqrange},
  freqrange = 1000 π;
  nyp = ParametricPlot[{Re[W /. s → ⅈ ω], Im[W /. s → ⅈ ω]},
    {ω, If[TrueQ[type == p], 0, -freqrange], freqrange}, Evaluate[
     FilterOptions[ParametricPlot, opts]], AxesLabel → {"Re", "Im"} ,
    DisplayFunction → Identity, PlotDivision → 50 freqrange];
  Show[GraphicsArray[{nyp}, DisplayFunction → $DisplayFunction],
   Frame → True, PlotLabel → "Nyquist - diagramm"];]
```

Nyquist diagrammot rajzol. Első két bemenő paramétere megegyezik a **BodePlot** parancséval. A harmadik paraméter az ábrázolás negatív frekvenciákra való kiterjesztését tiltja, vagy engedélyezi. Ezt követően további megjelenítésre vonatkozó utasítások adhatók meg. Lásd **Plot** utasítás. Fontos, hogy a második paraméter a frekvencia szerinti változó legyen, annak ellenére, hogy a Nyquist-diagramm másra is használható. Lásd később a Nyquist-diagramm bemutatásánál.

PoleZeroPlot

```
Options[PoleZeroPlot] = TextStyle →
   {FontFamily → "Helvetica", FontWeight → "Bold", FontSize → 12};

PoleZeroPlot[W_, s_, opts___?OptionQ] :=
  Show[Graphics[pzplace[{Re[#], Im[#]} & /@ Poles[W, s],
     {Re[#], Im[#]} & /@ Zeros[W, s]]], opts, Axes → True,
   AxesLabel → {"Re", "Im"}, AspectRatio → Automatic];
pzplace[poles_, zeros_] :=
  Append[#, First[Options[PoleZeroPlot]]] & /@ Flatten[
    {Text["o", #, {0, 0}] & /@ zeros, Text["x", #, {0, 0}] & /@ poles}];
```

Pólus-zérus elrendezés ábrázolására szolgáló parancs. Első két bemenő paramétere megegyezik a **BodePlot** parancséval. A harmadik paraméterek a megjelenítésre vonatkozó opciók lehetnek.

Frekvenciafüggés bemutatása

Tekintsük a következő egyszerű hálózatot, az integráló tagot.

```
ClearAll[layout1];
layout1 =
```

CornerUpperLeft	ResistorHorizontal[R]	NodeTDown	OutputHoriz
ACSourceVertical[u_1]	EmptyField	CapacitorVertical[C]	
CornerLowerLeft	WireHorizontal	NodeTUp	OutputHoriz

```
;
MakeDiagram[layout1]
```



- GraphicsArray -

A hálózatban két diszkrét komponens van, egy ellenállás és egy kondenzátor. A kondenzátor frekvenciafüggő alkotóelem, mert reaktanciája függ a frekvenciától. $X_c = \frac{1}{j\omega C}$. A két kapocs közti feszültség meghatározás a cél.

```
ClearAll[adatok];
adatok = {CC → 4.7, R → 39};
```

Az adatok valós alkatrészek adatai, a koherens egységrendszer μF és Ω. A hálózat egy aluláteresztő karakterisztikájú szűrő. Méretezése folytán a határfrekvencia≈880 Hz, vagyis 1 oktávval magasabb a normál zenei A hangtól.

■ Hálózati egyenletek

A hálózati egyenleteket frekvenciatartományba írjuk fel feszültségosztás módszerével. u_2 a kimeneten megjelenő feszültséget jelenti.

```
ClearAll[u1, u2];
u2 = Solve[(u2 - u1) / R + u2 / (1 / jω CC) == 0, u2][[1, 1, 2]]
```

$$\frac{\text{CC u1}}{\text{CC} + \text{j}\omega\text{ R}}$$

Az átviteli karakterszitka a kimenet és bemenet hányadosa adja meg. Az adatokat behelyettesítve megkapjuk a hálózat átviteli karakterisztikáját.

```
ClearAll[W];
W = u2 / u1 /. adatok
```

$$\frac{4.7}{4.7 + 39\text{ j}\omega}$$

Az átviteli karakterisztika egy függvény, aminek jω a változója, ahol j a komplex egység (i), ω a hálózatra kapcsolt jel körfrekvenciája.

$$H(\mathrm{j}\omega) = K\,\frac{N(\mathrm{j}\omega)}{D(\mathrm{j}\omega)} = |H(\mathrm{j}\omega)|\,e^{\mathrm{jArgH}(\mathrm{j}\omega)}$$

A képletben K a frekvenciától független rész.

Az átviteli karakterisztika ábrázolása

Ábrázolására két főbb eljárás terjedt el. Mindkettőt a Bell Laboratories-nál tevékenykedő tudósok dolgozták ki.

■ Bode - diagram

Az átviteli karakterisztika Bode-diagramja valójába két diagramm. Az egyik a decibelben kifejezett logaritmikus *amplitúdó karakterszitka*, másik a fokban kifejezett *fázis karakterszitka*. A diagramm $x-$tengelyéneka beosztása logaritmikus léptéket követ. Az $y-$tengelyfejezhet ki arányt (átviteli tényező), de átírható decibelre is dB = 20 $\text{Log}_{10}\left|W(j\,\omega)\right|$.

$$|H|_{\mathrm{dB}} = H_{\mathrm{dB}} = 20\log_{10}|H(\mathrm{j}\omega)|$$

A logaritmikus lépték előnye, hogy sokkal nagyobb függő és független változó tartomány lefedése lehetséges, ugyanakkora relatív hibával.

```
ClearAll[bodeexample];
bodeexample = 1 / ((s / ω)^2 + 2 (ξ / ω) s + 1) /. ω → 10
```

$$\frac{1}{1+\frac{s^2}{100}+\frac{s\,\xi}{5}}$$

A példában ξ különböző értékeire mutatjuk be a Bode-diagramot. $\xi \in \{0.05, 0.8\}$

```
BodePlot[Join[Table[bodeexample, {ξ, .05, .4, 0.05}],
  Table[bodeexample, {ξ, .4, .8, 0.1}]], s,
 FreqRange → {π, 10 π}, v, PlotStyle → {Hue[.125]}]
```



A példában szereplő hálózat átviteli karakterisztikáját ábrázolva.

A hálózat elsőrendű, frekvencia és fázismenete egyenessel közelíthető.

```
BodePlot[W /. adatok, jω, FreqRange → {0.002 π, 2 π},
 k, PlotRange → {Automatic, {0, 1}}]
```



- **Nyquist - diagramm**

A komplex síkon szemlélteti a $W(\mathrm{j}\omega)$ függvényt, a $H(\mathrm{j}\omega)$ végpontja által befutott utat szemlélteti a komplex számsíkon, miközben ω változik 0 vagy $-\infty$ és ∞ között. Elsősorban stabilitási kritériumok vizsgálatára használják, kvalitatív tájékozódásra alkalmas. Értelmezett negatív frekvenciára is. A negatív frekvenciákra felvett érték a pozitív komplex konjugáltja, azaz nem jelent többlet információt.

Előnye a Bode-diagramhoz képest, hogy nem korlátozott az ábrázolható frekvenciatartomány, ráadásul az amplitúdó és a fázis közös ábrán szerepelhet.

Nyquist - diagram akkor is használatos, ha az átviteli együttható értékét nem a körfrekvencia, hanem valamely más valós paraméter (pl. R rezisztancia) függvénye.

A példában egy konstans/harmadfokú alakú átviteli karakterisztikára, pozitív ω körfrekvenciákon mutatjuk be a Nyquist-diagramot. A diagramon jelöljük a pontok sűrűsödését a frekvencia növekedésével, valamint néhány pontnál a konkrét frekvenciaértéket is.

Látható, hogy a $\omega_1 = -100, \omega_2 = -200, \omega_3 = -300$ pólusok feletti frekvencián a pontok nagyon sűrűn helyezkednek.

```
ClearAll[nyquistexample];
```

$$\text{nyquistexample} = \frac{10^7}{(s+100)\,(s+200)\,(s+300)};$$

```
NyquistPlot[nyquistexample, s, p, PlotStyle → {Hue[.125]},
 Epilog → {PointSize[0.018], Hue[.125], (Point[{Re[#], Im[#]}] &) /@
    Join[Table[nyquistexample /. s → k ⅈ, {k, 1, 100, 10}],
     Table[nyquistexample /. s → k ⅈ, {k, 100, 200, 20}],
     Table[nyquistexample /. s → k ⅈ, {k, 200, 1000, 50}]]
   , (Text[#, {Re[nyquistexample /. s →  ⅈ #] + .1,
        Im[nyquistexample /. s →  ⅈ #] + .01}] &) /@
    Table[ω, {ω, 0, 300, 50}]}]
```



A példában szereplő hálózat átviteli karakterisztikáját Nyquist -diagrammon ábrázolva.

A hálózat elsőrendű. Ilyen hálózatok Nyquist -diagramja általában kör.

```
NyquistPlot[W /. adatok, jω, n,
  PlotRange → {{0, 1}, {-.5, .5}}, AspectRatio → 1]
```



Hálózatszámítás Mathematica-val

Időtartomány

A frekvenciatartományban végzett számítások elvégzése előtt időtartományban oldunk meg *dinamikus elemeket* tartalmazó hálózatra felírt egyenletet.

- **Dinamikus elemek**

A kondenzátor és tekercs tartozik ide, ezeket a továbbiakban lineáris és invariáns elemeknek tekintjük. A dinamikus elemeket állapotváltozóval jellemezük. Állapotváltozó a tekercs árama, és a kondenzátor feszültsége.

A kondenzátor olyan kétpólus, amelynek u_c feszültsége arányos a felvitt töltéssel, amely a i_c áram integrálja.

$$u_c = \frac{1}{C} \int_{-\infty}^{t} i_c(\tau)\, d\tau$$

A tekercs olyan kétpólus, amelynek i_L árama arányosa tekercs Φ mágneses fluxusával, amely a tekercs u_L feszültségének az integrálja.

$$i_L = \frac{1}{L} \int_{-\infty}^{t} u_L(\tau)\, d\tau$$

```
MakeDiagram[{CapacitorVertical[C], InductorHorizontal[L]}]
```



- GraphicsArray -

- **Hálózat**

```
ClearAll[layout2];
layout2 = {(*Elso sor*){EmptyField,
    EmptyField, CornerUpperLeft, WireHorizontal, NodeTDown,
    WireHorizontal, WireHorizontal, CornerUpperRight},
   (*Masodik sor*){EmptyField, EmptyField, ResistorVertical[R],
    EmptyField, CapacitorVertical[DisplayForm[SubscriptBox[C, 2]]],
    EmptyField, EmptyField, NodeTRight, CornerUpperRight},
   (*Harmadik sor*){CornerUpperLeft,
    CapacitorHorizontal[DisplayForm[SubscriptBox[C, 1]]],
    NodeTUp, ResistorHorizontal["10R"], NodeTUp, OPAmp1, OPAmp2,
    CornerLowerRight, ResistorVertical[DisplayForm[R / 2]]},
   (*Negyedik sor*){ACSourceVertical[
     DisplayForm[SubscriptBox[u, s]]], EmptyField, EmptyField,
    EmptyField, CornerUpperLeft, OPAmp3, OPAmp4, EmptyField,
    CapacitorVertical[DisplayForm[SubscriptBox[C, 3]]]},
   (*Ötödik sor*){CornerLowerLeft, WireHorizontal,
    WireHorizontal, WireHorizontal, NodeTUp, WireHorizontal,
    WireHorizontal, WireHorizontal, CornerLowerRight}};
MakeDiagram[layout2];
```



- **Szeretnénk kifejezni a hálózat impulzusválaszát. A válaszjel C_2 kondenzátor feszültsége.**

A válasz kifejezéséhez állapotegyenleteket kell felírni. Az állapotegyenlet az állapotváltozókra felírt differenciálegyenlet. Az állapotváltozók jelen esetben a kondenzátor feszültségek. Az egyenletek felírásához a csomóponti potenciálok módszerét alkalmazzuk, elsősorban a műveleti erősítő miatt.

Az adatok: $R = 75\,\Omega, C = 160\,\mathrm{nF}$

Koherens egységrendszer

A numerikus számítások megkönnyítése érdekében koherens mértékegységrendszerre térünk át.
$[R] = \text{k}\Omega,\ [C] = \mu\text{F}$
A τ időállandó mértékegysége: $[R]\times[C] = [\text{ms}]$

$R = 0.75\ \text{k}\Omega, C = 0.016\ \mu\text{F}$

- **Az egyenletekből a következő felírás adódik**

Mátrixos alakban: $\begin{pmatrix} u_{c1}' \\ u_{c2}' \\ u_{c3}' \end{pmatrix} = \begin{pmatrix} \frac{-11}{10\,R\,C} & \frac{1}{R\,C} & 0 \\ \frac{-1}{10\,R\,C} & 0 & 0 \\ 0 & \frac{2}{R\,C} & \frac{-2}{R\,C} \end{pmatrix} \begin{pmatrix} u_{c1} \\ u_{c2} \\ u_{c3} \end{pmatrix} + \begin{pmatrix} \frac{-11}{10\,R\,C} \\ \frac{-1}{10\,R\,C} \\ 0 \end{pmatrix} u_s$

A keresett mennyiség: u_{c2}
Erre külön egyenlet írható fel, ami mátrixos leírás második sorával egyezik meg.

$u_{c2} = \begin{pmatrix} 0 & 1 & 0 \end{pmatrix} \begin{pmatrix} u_{c1} \\ u_{c2} \\ u_{c3} \end{pmatrix} + 0\,u_s$

```
ClearAll[adatok];
adatok = {R → .75, CC → .016};
```

A rendszermátrix

```
ClearAll[A];
A = {{-11 / (10 R CC), 1 / (R CC) , 0},
    {-1 / (10 R CC), 0, 0},
    {0, 2 / (R CC) , -2 / (R CC)}} /. adatok;
```

```
A // MatrixForm
```

$$\begin{pmatrix} -91.6667 & 83.3333 & 0 \\ -8.33333 & 0 & 0 \\ 0 & 166.667 & -166.667 \end{pmatrix}$$

Stabilitási kritérium

A hálózat aszimptotikusan stabil, ha az $\boldsymbol{A}$ rendszermátrix minden sajátértékének a valós része negatív.

```
ClearAll[λ];
λ = Eigenvalues[A]
```

```
{-166.667, -83.3333, -8.33333}
```

```
If[And @@ Outer[#1 < 0 &, Re[λ]],
 CellPrint[Cell["Aszimptotikusan stabil.", "Text"]],
 CellPrint[Cell["Nem stabil!!!", "Text",
   CellFrame → True, Background → RGBColor[1, 0, 0]]]]
```

Aszimptotikusan stabil.

```
ClearAll[B];
B = {{-11 / (10 R CC), -1 / (10 R CC) , 0}} /. adatok;
```

```
B // MatrixForm
```

$$\begin{pmatrix} -91.6667 & -8.33333 & 0 \end{pmatrix}$$

■ Differenciálegyenlet megoldása

A differenciálegyenlet megoldása, úgy történik, hogy először a szabad összetevőt határozzuk meg. Ez matematikailag a homogén differenciálegyenlet megoldásának felel meg.

```
ClearAll[u1, u2, u3];
homogen = {uc1'[t] == -11 / (10 R CC) uc1[t] + 1 / (R CC) uc2[t] ,
    uc2'[t] == -1 / (10 R CC) uc1[t],
    uc3'[t] == 2 / (R CC) uc2[t] - 2 / (R CC) uc3[t]} /. adatok;
```

```
DSolve[homogen, {uc1[t], uc2[t], uc3[t]}, t]
```

$$\{\{uc1[t] \to e^{-258.333\,t}\,(1.11111\,e^{175.\,t} - 0.111111\,e^{250.\,t})\,C[1] + e^{-258.333\,t}\,(-1.11111\,e^{175.\,t} + 1.11111\,e^{250.\,t})\,C[2] + 0.\,e^{-258.333\,t}\,(e^{91.6667\,t} + e^{175.\,t} + e^{250.\,t})\,C[3],$$
$$uc2[t] \to e^{-258.333\,t}\,(0.111111\,e^{175.\,t} - 0.111111\,e^{250.\,t})\,C[1] + e^{-258.333\,t}\,(-0.111111\,e^{175.\,t} + 1.11111\,e^{250.\,t})\,C[2] + 0.\,e^{-258.333\,t}\,(e^{91.6667\,t} + e^{175.\,t} + e^{250.\,t})\,C[3],\ uc3[t] \to$$
$$e^{-258.333\,t}\,(-0.105263\,e^{91.6667\,t} + 0.222222\,e^{175.\,t} - 0.116959\,e^{250.\,t})\,C[1] + e^{-258.333\,t}\,(-0.947368\,e^{91.6667\,t} - 0.222222\,e^{175.\,t} + 1.16959\,e^{250.\,t})\,C[2] + 1.\,e^{-166.667\,t}\,C[3]\}\}$$

Implulzusválasz meghatározásánál, nincs gerjesztett összetevő, csak szabad, mert a Dirac-impulzus csak a $t > 0$ időtartományban gerjeszti a rendszert. Így a homogén egyenletből, és a kezdeti feltételekből megkapjuk a megoldást. Matematikailag az inhomogén egyenlet partikuláris megoldása 0.

A kezdeti értékek, vagyis az állapotváltozók értéke a $t = +0$ időpillanatban, b állapotváltozós egyenletben szereplő vektor lesz. Ekkor az állapotváltókban lehet ugrás, mivel a gerjesztés Dirac-impulzust tartalmaz.

```
homogenstate = {uc1'[t] == -11 / (10 R CC) uc1[t] + 1 / (R CC) uc2[t] ,
    uc2'[t] == -1 / (10 R CC) uc1[t],
    uc3'[t] == 2 / (R CC) uc2[t] - 2 / (R CC) uc3[t], uc1[0] ==
     -11 / (10 R CC), uc2[0] == -1 / (10 R CC), uc3[0] == 0} /. adatok;
```

```
homogenstate
```

{uc1′[t] == −91.6667 uc1[t] + 83.3333 uc2[t],
 uc2′[t] == −8.33333 uc1[t], uc3′[t] == 166.667 uc2[t] − 166.667 uc3[t],
 uc1[0] == −91.6667, uc2[0] == −8.33333, uc3[0] == 0}

```
DSolve[homogenstate, {uc1[t], uc2[t], uc3[t]}, t]
```

$\{\{uc1[t] \to e^{-258.333\,t}\,(0.\,e^{91.6667\,t} - 92.5926\,e^{175.\,t} + 0.925926\,e^{250.\,t}),$
$uc2[t] \to e^{-258.333\,t}\,(0.\,e^{91.6667\,t} - 9.25926\,e^{175.\,t} + 0.925926\,e^{250.\,t}),$
$uc3[t] \to e^{-425.\,t}\,(17.5439\,e^{258.333\,t} - 18.5185\,e^{341.667\,t} + 0.974659\,e^{416.667\,t})\}\}$

```
imp = %[[1, 2, 2]] // Simplify
```

$0.\,e^{-166.667\,t} - 9.25926\,e^{-83.3333\,t} + 0.925926\,e^{-8.33333\,t}$

```
Options[impvalasz] =
  {PlotStyle -> {Hue[.125]}, PlotLabel -> "-==Impulzusválasz==-",
   AxesLabel -> {"t", DisplayForm[SubscriptBox[u, c2]]}};
```

```
Plot[imp, {t, 0, 1}, Evaluate[Options[impvalasz]]]
```



- Graphics -

Frekvenciatartomány

■ **Most ugyanerre a hálózatra írunk fel frekvenciatartománybeli egyenleteket.**

Az egyenletek felírásához a csomóponti potenciálok módszerét alkalmazzuk.

```
MakeDiagram[layout2]
```




```
- GraphicsArray -
```

A műveleti erősíti invertáló bemenetére felírt egyenlet

$$\frac{-u_2}{\frac{1}{s C_2}} - \frac{\phi}{10 R} = 0$$

A ϕ potenciálú csomópontra felírt egyenlet

$$\frac{\phi - u_2}{\frac{1}{s C_1}} + \frac{\phi - u_2}{R} + \frac{\phi}{10 R} = 0$$

```
ClearAll[u1, u2, ϕ];
```

```
trans = {-u2 s CC - ϕ / (10 R) == 0, s CC (ϕ - u1) + (ϕ - u2) / R + ϕ / (10 R) == 0};
```

```
Solve[trans, u2, ϕ]
```

$$\left\{\left\{u2 \to -\frac{CC\ R\ s\ u1}{1 + 11\ CC\ R\ s + 10\ CC^2\ R^2\ s^2}\right\}\right\}$$

```
W = %[[1, 1, 2]] / u1
```

$$-\frac{CC\ R\ s}{1 + 11\ CC\ R\ s + 10\ CC^2\ R^2\ s^2}$$

■ **Az átviteli karakterisztika**

```
W /. adatok
```

$$-\frac{0.012\,s}{1+0.132\,s+0.00144\,s^2}$$

```
BodePlot[W /. adatok, s, FreqRange → Automatic,
 v, PlotStyle → {Hue[.125]}, PlotRange → {0, -90}]
```

$$\left\{\frac{16}{\pi^2},\ 168\,\pi\right\}$$

Bode – diagamm
Amplitúdó & Fázis



NyquistPlot[W /. adatok, s, ℂ, AspectRatio → 1]



PoleZeroPlot[W /. adatok, s, AspectRatio → 1]



- Graphics -

- **A hálózat impulzusválaszának meghatározása**

Az impulzusválasz időfüggvénye az átviteli karakterisztika inverz Laplace-transzformálásával kapható meg.

```
wf[t_] = InverseLaplaceTransform[W, s, t] /. adatok // Simplify
```

$e^{-83.3333\,t}\,(-9.25926 + 0.925926\,e^{75.\,t})$

```
Plot[wf[t], {t, 0, 1}, Evaluate[Options[impvalasz]]]
```



- Graphics -

Diszkrét szimuláció

Diszkrét idejű hálózatok

A diszkrét idő a $t_k = \mathrm{kT}$, $k \in \mathbb{Z}$ időpillanatokat jelenti. Diszkrét idejű jelek esetén ezekhez az időpillanatokhoz rendelünk valós értékeket.
A diszkrét idejű hálózat változói diszkrét idejűek.

Diszkrét idejű hálózatok komponensei

A hálózat kevés elemmel leírható. Be- és kimeneten kívül összegző csomópont, késleltető és szorzó fordul elő.

```
MakeDiscreteDiagram[{MultiplierLeft["Const"],
  Graphics[{}], DelayLeft, Graphics[{}], Adder2WayLeft}]
```




```
- GraphicsArray -
```

Karakterisztikák
Szorzó $v[k] = c\, u[k]$
Késleltető $v[k] = u[k-1]$
Összegző $v[k] = \sum_{i=1}^{n} u_i[k]$

A *szimuláció*-t a következő értelemben használjuk: Adott egy folytonos idejű rendszer, feladat egy olyan diszkrét idejű rendszer meghatározása, amelynek viselkedése "hasonló" a folytonos idejű megfelelőjéhez. A szimuláció tekinthető egy numerikus közelítésnek is. A szimuláló diszkrét idejű hálózat számítástechnikailag előnyösebb.

A diszkrét szimulációhoz nem csak a hálózat szimulálását jelenti, hanem a jelek diszkrét idejűvé alakítását is. A diszkrét jelhez mintavételezésre van szükség. Válasszunk T mintavételezési időt.
A szimuláció ideális esetben azt eredményezi, hogy a szimuláló rendszer válasza megegyezik a folytonos idejű rendszer válaszának mintáival, vagyis

$$u_d[k] = u_c(\mathrm{kT}) \to y_d[k] = y_c(\mathrm{kT})$$

teljesül bármilyen $u_{cc}(t)$ gerjesztés esetén.

A szimulátor megalkotásakor a cél az optimalizálás, azaz minél egyszerűbb rendszerrel érjünk el minél nagyobb pontosságot. A szimuláció pontossága azonos feltételek mellett akkor pontosabb, ha a T mintavételi idő minél kisebb.

Átviteli karakterisztika szimulálása

Tekintsük adottnak a folytonos idejű rendszer $H_c(s)$ átviteli függvényét. A cél, hogy a szimuláló rendszer gerjesztett válasza megegyezzen a folytonos idejű rendszer gerjesztett válaszának mintáival.

$$u_c(t) = e^{i\,\omega\,t},\; u_d[k] = e^{i\,\omega\,k\,T} \to u_d[k] = e^{i\,\omega\,k\,T}\,\big|_{\theta=\omega\,T}$$

A válasz gerjesztett összetevőinek komplex alakja

$$y_c(t) = H_c(j\,\omega)\, e^{i\,\omega\,t},\; y_d[k] = H_d(e^{i\,\omega\,T})\, e^{i\,\omega\,k\,T}$$

A szimuláció ideális, ha $y_d[k] = y_c(\text{kT})$ feltétel teljesül. Ennek feltétele

$$H_d(e^{i\,\theta}) = H_c(j\,\omega)\,|_{\omega=\theta/T} \rightarrow H_d(z) = H_c(s)\,|_{s=T^{-1}\ln z}$$

Ez azonban nem teljesíti azt a feltétel, hogy az átviteli függvény racionális legyen. Ez pedig nem realizálható az ismertetett elemekkel.

Ezzel a tulajdonsággal a bilineáris transzformáció rendelkezik. A képletben szereplő p paraméter szokásos választása $p = 2$. A transzformáció megtartja továbbá az átviteli függvény pólusainak és zérusainak kapcsolatát.

$$H_d(z) = H_c(s)\,|_{s=f(z)},\; f(z) = \frac{p}{T}\,\frac{z-1}{z+1}\quad 0 < p \leq 2$$

Most egy 22050 Hz mintavételezési mintával szeretnénk dolgozni, de mivel a koherens egységrendszerben ms -ban számoltunk T helyére 22.05-öt írunk. Mivel ismerjük a folytonos idejű rendszer átviteli karakterisztikáját a bilineáris transzformációval meghatározható a diszkrét hálózaté.

```
ClearAll[Wd];
Wd = (W /. s → 2 / 22.05 (z - 1) / (z + 1) // Together) /. adatok
```

$$-\frac{0.00108844\,(-1.+z)\,(1.+z)}{0.988039+1.99998\,z+1.01198\,z^2}$$

A diszkrét hálózat esetén a hálózatra értelmezhet amplitúdó- és fáziskarakterisztika. Az amplitúdó-karakterisztika $e^{i\theta}$ miatt páros függvény, míg a fázis-karakteriszitka páratlan. Mindkettő 2π szerint periodikus.

```
Plot[Abs[ExpToTrig[Wd /. z → ⅇ^(ⅈ θ)]], {θ, -2 π - π / 4, 2 π + π / 4},
 Ticks → {Range[-2 π, 2 π, π / 2], Automatic}, AspectRatio → 1 / 2]
```



- Graphics -

```
Plot[Arg[ExpToTrig[Wd /. z → ⅇ^(ⅈ θ)]], {θ, -2 π - π / 4, 2 π + π / 4},
 Ticks → {Range[-2 π, 2 π, π / 2], Automatic}, AspectRatio → 1 / 2]
```



- Graphics -

```
Show[PoleZeroPlot[Wd, z, DisplayFunction → Identity],
 Graphics[Circle[{0, 0}, 1]], DisplayFunction → $DisplayFunction,
 PlotRange → {{-1.5, 1.5}, {-1.5, 1.5}}]
```



- Graphics -

Impulzusválasz szimulálása

Tekintsük most adottnak a folytonos idejű rendszer impulzusválaszának időfüggvényét az alábbi alakban

$$h_c(t) = \mathrm{D}\delta(t) + \epsilon(t)\, f(t)$$

Ekkor a hálózat válasza tetszőleges $u_c(t)$ belépő gerjesztésre konvolúcióval határozható meg

$$y_c(t) = \int_{-0}^{t} h_c(\tau)\, u_c(t-\tau)\, d\tau$$

Az integrál közelítése diszkrét időben

$$y_d[k] = \sum_{i=0}^{k} h_d[i]\, u_d[k-i],\ k\in\mathbb{N} \quad \text{vagy másképp}$$
$$y_d[k] = \sum_{i=0}^{k} h_d[k-i]\, u_d[i]$$

Ez alapján a következő szimulációs szabályt kapjuk

$$h_c(t) = \mathrm{D}\delta(t) + \epsilon(t)\, f(t) \to h_d[k] = \mathrm{D}\delta[k] + \mathrm{T}\epsilon[k-1]\, f(k\,T)$$

```
mintavetel = {D → B⟦1, 2⟧, T → 1 / 22.05};
```

```
ClearAll[wd];
wd[k_] :=
 wd[k] = D DiscreteDelta[k] + T UnitStep[k - 1] wf[k T] /. mintavetel
```

Most a mintavételezett impulzusválaszt szeretnénk szemléltetni, vizsgálva, hogy valóban megegyez-ik-e a mintavételezett jellel.. A *Mathematica* Spline függvényével görbét illeszthetünk a diszkrét értékekre.

```
<< Graphics`Spline`
```

```
ClearAll[ertekek];
ertekek = Table[{k, wd[k]}, {k, 0, 10}];
```

```
Show[{ListPlot[ertekek, PlotStyle → {Hue[1]},
   DisplayFunction → Identity], Graphics[Spline[ertekek,
    Cubic, SplineDots → {RGBColor[0, 0, 0], PointSize[.02]}]],
  Graphics[Spline[ertekek, Bezier]]},
 DisplayFunction → $DisplayFunction]
```




```
- Graphics -
```

```
ListPlot[Table[wd[k], {k, 0, 100}]]
```




```
- Graphics -
```

Szimulált válasz

Most következünk a szimuláció. A hálózatra hangfrekvenciás mintavételezett jelet kapcsolunk. A jel adatai az alábbi táblázatból olvashatók ki.

```
<< Miscellaneous`Audio`
```

```
snd = ReadSoundFile["C:\startm.wav", PrintHeader → True];
```

```
Format: Microsoft PCM WAVE RIFF

Duration: 0.0458957 seconds

Channels: 1

Sampling rate: 22050

Bits per sample: 16

Data size: 2024 bytes

Number of samples: 1012
```

```
ListPlay[snd, SampleRate → 22050]
```



```
- Sound -
```

Írjuk fel a kétféle diszkrét konvolúció képletet. Majd vizsgáljuk meg melyik az előnyösebb a feladat szempontjából.

```
ClearAll[y1];
```

$$\text{y1}[\text{k\_}] := \sum_{n=0}^{k} \text{snd}[\![k - n + 1]\!] * \text{wd}[n]$$

```
ClearAll[y2];
```

$$\text{y2}[\text{k\_}] := \sum_{n=0}^{k} \text{snd}[\![n + 1]\!] * \text{wd}[k - n]$$

```
Timing[Table[y1[k], {k, 0, 400}];](*1.59 s*)
```

```
{1.81 Second, Null}
```

```
Timing[Table[y2[k], {k, 0, 400}];](*1.38 s*)
```

```
{1.48 Second, Null}
```

```
ListPlay[Table[y2[k], {k, 0, Length[snd] - 1}], SampleRate → 22050]
```



```
- Sound -
```

Látható hogy, az időtartományban végzett szimuláció idő- és számításigényessége.
A megoldás a frekvenciatartományra való áttérés. Fourier transzformáció, Diszkrét Fourier Transzformáció (DFT), gyors Fourier transzformáció (FFT), diszkrét koszinusz transzformáció, stb. alkalmazása.

Felhasznált irodalom

Fodor György: Hálózatok és rendszerek analízise

Fodor György: Jelek, rendszerek, hálózatok

Gyimes László: Digitális jelfeldolgozás

The Electrical Engeneering Handbook

Mathematica Information Center